# $Id: PKGBUILD 158778 2012-05-09 17:40:49Z ibiru $
# Maintainer : Ionut Biru <ibiru@archlinux.org>

pkgname=telepathy-rakia
pkgver=0.7.4
pkgrel=1
pkgdesc="SIP connection manager for Telepathy"
arch=('i686' 'x86_64')
url="http://telepathy.freedesktop.org"
license=('LGPL')
depends=('telepathy-glib' 'sofia-sip')
makedepends=('libxslt' 'python2')
install=telepathy-rakia.install
groups=('telepathy')
replaces=('telepathy-sofiasip')
source=(http://telepathy.freedesktop.org/releases/$pkgname/$pkgname-$pkgver.tar.gz
        0001-Check-for-gio-to-avoid-linking-issue.patch)
md5sums=('f9f47f18db033afe29fd6963c7748e37'
         '6d0372c4c8a22bf9682e4b45ee614d5c')

build() {
  cd "$pkgname-$pkgver"
  patch -Np1 -i $srcdir/0001-Check-for-gio-to-avoid-linking-issue.patch
  autoreconf -fi
  ./configure --prefix=/usr --libexecdir=/usr/lib/telepathy
  make
}

package() {
  cd "$pkgname-$pkgver"
  make DESTDIR="$pkgdir" install
}
